Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

# Lesson 10
## Motion in Space

# Initializations

```
> restart;
  with(VectorCalculus):
  BasisFormat(false):
>
```

# 10.1 Motion in Space

## Examples

### *Example 10.1.1*

Suppose a projectile is shot out of a barrel under an angle of 60 degrees with the horizontal and with a speed of 2000 ft/sec. Find the position of the projectile as a function of the time *t*.

**Solution**

Of course the projectile is subject to gravity, which results in an acceleration vector **a** = $\langle 0, -32 \rangle$. We enter the data. Assume the barrel is located in the origin.

```
> v0:=2000*<cos(Pi/3), sin(Pi/3)>:
  r0:=<0, 0>:
  a:=<0, -32>:
  v0, r0, a;
```

$$\begin{bmatrix} 1000 \\ 1000\sqrt{3} \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ -32 \end{bmatrix} \quad \textbf{(2.1.1.1)}$$

Now define a vector **C** which will serve as an integration constant and integrate $\mathbf{a}(t)$.

```
> C:=<c[1], c[2]>;
  ai:=int(a, t)+C;
```

$$C := \begin{bmatrix} c_1 \\ c_2 \end{bmatrix}$$

$$ai := \begin{bmatrix} c_1 \\ -32\,t + c_2 \end{bmatrix} \quad \textbf{(2.1.1.2)}$$

Impose the initial condition $\mathbf{v}(0) = \mathbf{v}_0$, and compute $\mathbf{v}(t)$.

```
> val_c:=solve(Equate(subs(t=0, ai), v0), {c[1], c[2]});
```

```
v:=subs(val_c, ai);
```

$$val\_c := \{c_1 = 1000, c_2 = 1000\sqrt{3}\}$$

$$v := \begin{bmatrix} 1000 \\ -32\,t + 1000\sqrt{3} \end{bmatrix} \quad \textbf{(2.1.1.3)}$$

Integrate this result and implement the initial condition $\mathbf{r}(0) = \mathbf{r}_0$ to find $\mathbf{r}(t)$.

```
> vi:=int(v, t)+C;
```

$$vi := \begin{bmatrix} 1000\,t + c_1 \\ -16\,t^2 + 1000\sqrt{3}\,t + c_2 \end{bmatrix} \quad \textbf{(2.1.1.4)}$$

```
> val_c:=solve(Equate(subs(t=0, vi), r0), {c[1], c[2]});
  r:=subs(val_c, vi);
```

$$val\_c := \{c_1 = 0, c_2 = 0\}$$

$$r := \begin{bmatrix} 1000\,t \\ -16\,t^2 + 1000\sqrt{3}\,t \end{bmatrix} \quad \textbf{(2.1.1.5)}$$

It is possible to sketch the path of the projectile using a parametric plot.

```
> plot([r[1], r[2], t=0..100]);
```




# 10.2 Tangential and Normal Components of the Acceleration

## Examples

### *Example 10.2.1*

Suppose the location of a point-mass is given by $r(t) = \langle t, t^2, t^3 \rangle$. Compute the tangential and normal components of the acceleration.

**Solution**

Enter $\mathbf{r}(t)$ and compute its first and second derivative.

```
> r:=<t, t^2, t^3>:
  rp:=diff(r, t):
  rpp:=diff(r, t$2):
  r, rp, rpp;
```

$$\begin{bmatrix} t \\ t^2 \\ t^3 \end{bmatrix}, \begin{bmatrix} 1 \\ 2\,t \\ 3\,t^2 \end{bmatrix}, \begin{bmatrix} 0 \\ 2 \\ 6\,t \end{bmatrix} \qquad \textbf{(3.1.1.1)}$$

Both $a_T$ and $a_N$ can now readily be computed using the formulas derived in-class. We clear the variable **a** since it has been used in the previous example.

```
> a:='a';
  a[T]:=rp.rpp/Norm(rp);
  a[N]:=Norm(rp &x rpp)/Norm(rp);
```

$$a := a$$

$$a_T := \frac{4\,t + 18\,t^3}{\sqrt{1 + 4\,t^2 + 9\,t^4}}$$

$$a_N := \frac{2\sqrt{1 + 9\,t^4 + 9\,t^2}}{\sqrt{1 + 4\,t^2 + 9\,t^4}} \qquad \textbf{(3.1.1.2)}$$

```
>
```